



# Security Audit

# Report for iBTC

# Contracts

**Date:** November 22, 2024 **Version:** 1.0

**Contact:** [contact@blocksec.com](mailto:contact@blocksec.com)

# Contents

<b>Chapter 1 Introduction</b>	<b>1</b>
1.1 About Target Contracts . . . . .	1
1.2 Disclaimer . . . . .	1
1.3 Procedure of Auditing . . . . .	2
1.3.1 Software Security . . . . .	2
1.3.2 DeFi Security . . . . .	2
1.3.3 NFT Security . . . . .	3
1.3.4 Additional Recommendation . . . . .	3
1.4 Security Model . . . . .	3
<b>Chapter 2 Findings</b>	<b>5</b>
2.1 Software Security . . . . .	5
2.1.1 Flawed sorting logic in the <code>updateValidator</code> function . . . . .	5
2.2 DeFi Security . . . . .	6
2.2.1 Flawed reward accounting logic in the <code>iBTC</code> contract . . . . .	6
2.2.2 Flawed reward distribution logic due to split atomic operations . . . . .	8
2.3 Additional Recommendation . . . . .	9
2.3.1 Traverse the <code>validators</code> array from the end when removing a validator . . . . .	9
2.3.2 Avoid using <code>address.transfer</code> for transferring native tokens . . . . .	10
2.3.3 Remove redundant operations in the <code>iBTC</code> contract . . . . .	11
2.3.4 Optimize the sorting algorithm to reduce gas consumption . . . . .	13
2.4 Note . . . . .	14
2.4.1 Potential centralization risk . . . . .	14

## Report Manifest

Item	Description
Client	KikiFinance
Target	iBTC Contracts

## Version History

Version	Date	Description
1.0	November 22, 2024	First release

## Signature

**About BlockSec** BlockSec focuses on the security of the blockchain ecosystem and collaborates with leading DeFi projects to secure their products. BlockSec is founded by top-notch security researchers and experienced experts from both academia and industry. They have published multiple blockchain security papers in prestigious conferences, reported several zero-day attacks of DeFi applications, and successfully protected digital assets that are worth more than 14 million dollars by blocking multiple attacks. They can be reached at [Email](#), [Twitter](#) and [Medium](#).

# Chapter 1 Introduction

## 1.1 About Target Contracts

Information	Description
Type	Smart Contract
Language	Solidity
Approach	Semi-automatic and manual verification

The focus of this audit is on the iBTC Contracts <sup>1</sup> of KikiFinance. The iBTC Contracts enable users to deposit **xBTC** in exchange for **iBTC** on a 1:1 basis and delegate the staked **xBTC** to multiple validators via **StakeRouter** to earn rewards (typically **XSAT**), which are then distributed to **iBTC** holders.

Please note that the audit scope is limited to the contracts in the **contracts** folder. Other files are not within the scope of the audit. Additionally, all dependencies of the smart contracts within the audit scope are considered reliable in terms of both functionality and security, and are therefore not included in the audit scope.

The auditing process is iterative. Specifically, we would audit the commits that fix the discovered issues. If there are new issues, we will continue this process. The commit SHA values during the audit are shown in the following table. Our audit report is responsible for the code in the initial version (**Version 1**), as well as new code (in the following versions) to fix issues in the audit report.

Project	Version	Commit Hash
iBTC Contracts	<b>Version 1</b>	3182d5ed920e3ce171ac105a66745bf51eb94980
	<b>Version 2</b>	816ee4eb61fb72e29c8237f6a83e561f6ea13310
	<b>Version 3</b>	8adb47dbcd5700274808d6962a87ca18e6cda34f

The contracts are deployed on the exSat network using the same **Version 3** code. The deployed contract addresses are listed in the following table.

Contract		Address
StakeRouter	Proxy	0xfe34D8B434324479Bb2850864e9B2966765B81Be
	Impl	0x76058B3c4e1213A34FF884324b1a8E8678269237
iBTC	Proxy	0x8154Aaf094c2f03Ad550B6890E1d4264B5DdaD9A
	Impl	0xD090dC07d6679abf80BfF79a4662061098f47ec
WIBTC	-	0xC237fd3a90400082aD729fCc1cD88D179cb86e2b

## 1.2 Disclaimer

This audit report does not constitute investment advice or a personal recommendation. It does not consider, and should not be interpreted as considering or having any bearing on,

<sup>1</sup><https://github.com/KikiFinance/ibtc-contract/>

the potential economics of a token, token sale or any other product, service or other asset. Any entity should not rely on this report in any way, including for the purpose of making any decisions to buy or sell any token, product, service or other asset.

This audit report is not an endorsement of any particular project or team, and the report does not guarantee the security of any particular project. This audit does not give any warranties on discovering all security issues of the smart contracts, i.e., the evaluation result does not guarantee the nonexistence of any further findings of security issues. As one audit cannot be considered comprehensive, we always recommend proceeding with independent audits and a public bug bounty program to ensure the security of smart contracts.

The scope of this audit is limited to the code mentioned in Section 1.1. Unless explicitly specified, the security of the language itself (e.g., the solidity language), the underlying compiling toolchain and the computing infrastructure are out of the scope.

## 1.3 Procedure of Auditing

We perform the audit according to the following procedure.

- **Vulnerability Detection** We first scan smart contracts with automatic code analyzers, and then manually verify (reject or confirm) the issues reported by them.
- **Semantic Analysis** We study the business logic of smart contracts and conduct further investigation on the possible vulnerabilities using an automatic fuzzing tool (developed by our research team). We also manually analyze possible attack scenarios with independent auditors to cross-check the result.
- **Recommendation** We provide some useful advice to developers from the perspective of good programming practice, including gas optimization, code style, and etc.

We show the main concrete checkpoints in the following.

### 1.3.1 Software Security

- \* Reentrancy
- \* DoS
- \* Access control
- \* Data handling and data flow
- \* Exception handling
- \* Untrusted external call and control flow
- \* Initialization consistency
- \* Events operation
- \* Error-prone randomness
- \* Improper use of the proxy system

### 1.3.2 DeFi Security

- \* Semantic consistency
- \* Functionality consistency
- \* Permission management

- \* Business logic
- \* Token operation
- \* Emergency mechanism
- \* Oracle security
- \* Whitelist and blacklist
- \* Economic impact
- \* Batch transfer

### 1.3.3 NFT Security

- \* Duplicated item
- \* Verification of the token receiver
- \* Off-chain metadata security

### 1.3.4 Additional Recommendation

- \* Gas optimization
- \* Code quality and style



**Note** The previous checkpoints are the main ones. We may use more checkpoints during the auditing process according to the functionality of the project.

## 1.4 Security Model

To evaluate the risk, we follow the standards or suggestions that are widely adopted by both industry and academy, including OWASP Risk Rating Methodology <sup>2</sup> and Common Weakness Enumeration <sup>3</sup>. The overall *severity* of the risk is determined by *likelihood* and *impact*. Specifically, likelihood is used to estimate how likely a particular vulnerability can be uncovered and exploited by an attacker, while impact is used to measure the consequences of a successful exploit.

**Table 1.1:** Vulnerability Severity Classification

Impact	High	Low
	High	Low
Likelihood	High	Low
	High	Low

<sup>2</sup>[https://owasp.org/www-community/OWASP\\_Risk\\_Rating\\_Methodology](https://owasp.org/www-community/OWASP_Risk_Rating_Methodology)

<sup>3</sup><https://cwe.mitre.org/>

In this report, both likelihood and impact are categorized into two ratings, i.e., *high* and *low* respectively, and their combinations are shown in Table 1.1.

Accordingly, the severity measured in this report are classified into three categories: **High, Medium, Low**. For the sake of completeness, **Undetermined** is also used to cover circumstances when the risk cannot be well determined.

Furthermore, the status of a discovered item will fall into one of the following four categories:

- **Undetermined** No response yet.
- **Acknowledged** The item has been received by the client, but not confirmed yet.
- **Confirmed** The item has been recognized by the client, but not fixed yet.
- **Fixed** The item has been confirmed and fixed by the client.

## Chapter 2 Findings

In total, we found **three** potential security issues. Besides, we have **four** recommendations and **one** note.

- High Risk: 2
- Low Risk: 1
- Recommendation: 4
- Note: 1

ID	Severity	Description	Category	Status
1	Low	Flawed sorting logic in the <code>updateValidator</code> function	Software Security	Fixed
2	High	Flawed reward accounting logic in the <code>iBTC</code> contract	DeFi Security	Fixed
3	High	Flawed reward distribution logic due to split atomic operations	DeFi Security	Fixed
4	-	Traverse the <code>validators</code> array from the end when removing a validator	Recommendation	Fixed
5	-	Avoid using <code>address.transfer</code> for transferring native tokens	Recommendation	Fixed
6	-	Remove redundant operations in the <code>iBTC</code> contract	Recommendation	Fixed
7	-	Optimize the sorting algorithm to reduce gas consumption	Recommendation	Fixed
8	-	Potential centralization risk	Note	-

The details are provided in the following sections.

### 2.1 Software Security

#### 2.1.1 Flawed sorting logic in the `updateValidator` function

**Severity** Low

**Status** Fixed in [Version 2](#)

**Introduced by** [Version 1](#)

**Description** In the `StakeRouter` contract, the `updateValidator` function allows operators to update the configuration of an existing validator and sort the `validators` array by priority. However, the sorting logic is flawed. For example, if there are four validators with priorities 50, 30, 20, and 10, and the priority of the second validator is updated to 5, the `validators` array are reordered to priorities 50, 5, 20, and 10, resulting in an evidently incorrect order.

```
103 function updateValidator(  
104     address _validator,  
105     uint256 _minStakePerTx,  
106     uint256 _maxStake,  
107     uint256 _priority  
108 ) external onlyOperator {
```



```
109     require(_minStakePerTx <= _maxStake, "Minimum stake must be less than maximum stake");
110
111     int256 index = getValidatorIndex(_validator);
112     require(index >= 0, "Validator not found");
113
114     Validator storage validator = validators[uint256(index)];
115
116     if (validator.currentStake > 0) {
117         require(
118             validator.currentStake <= _maxStake,
119             "Current stake must be less than or equal to the maximum stake"
120         );
121     }
122
123     validator.minStakePerTx = _minStakePerTx;
124     validator.maxStake = _maxStake;
125     validator.priority = _priority;
126     emit ValidatorUpdated(_validator, _minStakePerTx, _maxStake, _priority);
127     _sortValidatorsByPriority(uint256(index));
128 }
```

**Listing 2.1:** StakeRouter.sol

```
74 function _sortValidatorsByPriority(uint256 startIndex) internal {
75     for (uint256 i = startIndex; i > 0; i--) {
76         if (validators[i].priority > validators[i - 1].priority) {
77             Validator memory temp = validators[i];
78             validators[i] = validators[i - 1];
79             validators[i - 1] = temp;
80         } else {
81             break;
82         }
83     }
84 }
```

**Listing 2.2:** StakeRouter.sol

**Impact** The `validators` array is incorrectly sorted.

**Suggestion** Revise the code logic accordingly.

## 2.2 DeFi Security

### 2.2.1 Flawed reward accounting logic in the iBTC contract

**Severity** High

**Status** Fixed in [Version 2](#)

**Introduced by** [Version 1](#)

**Description** In the `iBTC` contract, the `transfer` and `transferFrom` functions invoke the hook function `_beforeTokenTransfer` to settle rewards for both the sender and the receiver (i.e., the `from` and `to` addresses). However, after updating token balances, the contract fails to update

`rewardDebt`, which is critical for calculating pending rewards. This flaw leads to improper accounting, allowing an attacker to arbitrarily claim rewards through transfers.

For example, an attacker transfers 100 `iBTC` tokens to a controlled account with no token balance. The `rewardDebt` for the controlled account is set to zero in `_beforeTokenTransfer` before the balance update. The controlled account can then immediately invoke the `claimReward` function to claim XSAT tokens amounting to  $100 * \text{accRewardPerShare}$ , violating the intended design.

```
77 function _beforeTokenTransfer(address from, address to, uint256 amount) internal override {
78     if (from != address(0)) {
79         _settleReward(from);
80     }
81     if (to != address(0)) {
82         _settleReward(to);
83     }
84     super._beforeTokenTransfer(from, to, amount);
85 }
```

Listing 2.3: `iBTC.sol`

```
65 function _settleReward(address userAddress) internal {
66     require(xsatBalanceBefore == 0, "distributed rewards not yet finish");
67     uint256 pending = _pendingReward(userAddress);
68     if (pending > 0) {
69         xsat.safeTransfer(userAddress, pending); // Transfer pending XSAT reward
70         emit ClaimReward(userAddress, pending);
71     }
72     // Update the user's reward debt to reflect the latest accumulated reward per share
73     rewardDebt[userAddress] = (balanceOf(userAddress) * accRewardPerShare) / PRECISION;
74 }
```

Listing 2.4: `iBTC.sol`

```
55 function _pendingReward(address userAddress) internal view returns (uint256) {
56     uint256 userBalance = balanceOf(userAddress);
57     uint256 accumulatedReward = (userBalance * accRewardPerShare) / PRECISION;
58     if (accumulatedReward > rewardDebt[userAddress]) {
59         return accumulatedReward - rewardDebt[userAddress];
60     }
61     return 0;
62 }
```

Listing 2.5: `iBTC.sol`

```
223function claimReward() external nonReentrant {
224    _settleReward(msg.sender);
225}
```

Listing 2.6: `iBTC.sol`

**Impact** The attacker can arbitrarily claim rewards by transferring `iBTC` tokens to new addresses.

**Suggestion** Update the `rewardDebt` timely after the balance change.

## 2.2.2 Flawed reward distribution logic due to split atomic operations

**Severity** High

**Status** Fixed in [Version 2](#)

**Introduced by** [Version 1](#)

**Description** In the `iBTC` contract, rewards are accumulated in the `StakeRouter` contract and distributed to `iBTC` when the `finalizeRewardDistribution` function of the `StakeRouter` contract is invoked. The `iBTC` contract records its `XSAT` balance (stored in the `xsatBalanceBefore` variable) in the `prepareRewardDistribution` function, and calculates the amount of `XSAT` tokens received (based on the `xsatBalanceAfter` variable) in the `finalizeRewardDistribution` function after calling the corresponding function in the `StakeRouter` contract. However, these two operations should be atomic but are not in the current implementation, enabling an attacker to arbitrarily increase the reward per share in the `iBTC` contract.

Specifically, if the `finalizeRewardDistribution` function of the `StakeRouter` contract is executed successfully, the attacker can manipulate the `iBTC` contract's `XSAT` balance to be considered as rewards. Since the `finalizeRewardDistribution` function of the `StakeRouter` contract only requires its `XSAT` balance to be greater than zero, the attacker can transfer 1 wei of `XSAT` tokens to the `StakeRouter` contract to bypass this requirement. Consequently, the attacker can repeatedly invoke the `finalizeRewardDistribution` function of the `iBTC` contract to infinitely increase `accRewardPerShare`.

```

228  function prepareRewardDistribution() external nonReentrant {
229      _prepareRewardDistribution();
230  }
231
232  // Finalizes reward distribution by updating accumulated rewards and recalculating rewards per
    share
233  function finalizeRewardDistribution() external nonReentrant {
234      uint256 supply = totalSupply();
235      require(supply > 0, "No iBTC in circulation for reward distribution");
236      stakeRouter.finalizeRewardDistribution();
237
238      // Get the updated XSAT balance after reward distribution
239      uint256 xsatBalanceAfter = xsat.balanceOf(address(this));
240      require(xsatBalanceAfter >= xsatBalanceBefore, "Balance error: Insufficient XSAT");
241
242      // Calculate the newly received rewards and reset the balance tracker
243      uint256 amount = xsatBalanceAfter - xsatBalanceBefore;
244      xsatBalanceBefore = 0;
245
246      // Update the accumulated reward per share for iBTC holders
247      accRewardPerShare += (amount * PRECISION) / supply;
248
249      emit RewardDistributed(amount);
250  }

```

**Listing 2.7:** `iBTC.sol`

```

55  function _pendingReward(address userAddress) internal view returns (uint256) {

```

```
56     uint256 userBalance = balanceOf(userAddress);
57     uint256 accumulatedReward = (userBalance * accRewardPerShare) / PRECISION;
58     if (accumulatedReward > rewardDebt[userAddress]) {
59         return accumulatedReward - rewardDebt[userAddress];
60     }
61     return 0;
62 }
```

**Listing 2.8:** iBTC.sol

```
220 function finalizeRewardDistribution() external onlyIBTC nonReentrant {
221     // transfer all xsat reward to iBTC
222     uint256 amount = xsat.balanceOf(address(this));
223     require(amount > 0, "Balance error: Insufficient XSAT");
224
225     xsat.safeTransfer(address(iBTC), amount);
226     emit RewardDistributed(amount);
227 }
```

**Listing 2.9:** StakeRouter.sol

**Impact** The attacker can infinitely inflate the `accRewardPerShare` value, enabling them to drain all reward tokens.

**Suggestion** Revise the code logic accordingly.

## 2.3 Additional Recommendation

### 2.3.1 Traverse the `validators` array from the end when removing a validator

**Status** Fixed by [Version 2](#)

**Introduced by** [Version 1](#)

**Description** In the `StakeRouter` contract, the `removeValidator` function allows the operator to remove a validator whose `currentStake` is zero. According to the protocol's design, validators with smaller indexes have higher priority, so validators selected for removal typically have larger indexes and lower priority. However, the `getValidatorIndex` function traverses the `validators` array starting from index 0 to locate a validator with zero staked funds. This approach can fail due to gas exhaustion if the array is too large.

```
130 function removeValidator(address _validator) external onlyOperator {
131     int256 index = getValidatorIndex(_validator);
132     require(index >= 0, "Validator not found");
133
134     uint256 validatorIndex = uint256(index);
135     Validator storage validatorToRemove = validators[validatorIndex];
136
137     require(
138         validatorToRemove.currentStake == 0,
139         "Cannot remove a validator with an active stake"
140     );
141 }
```

```
142     // Shift elements to the left to maintain order
143     for (uint256 i = validatorIndex; i < validators.length - 1; i++) {
144         validators[i] = validators[i + 1];
145     }
146
147     // Remove the last element
148     validators.pop();
149     emit ValidatorRemoved(_validator);
150 }
```

**Listing 2.10:** StakeRouter.sol

```
257 function getValidatorIndex(address _validator) internal view returns (int256) {
258     for (uint256 i = 0; i < validators.length; i++) {
259         if (validators[i].validatorAddress == _validator) {
260             return int256(i);
261         }
262     }
263     return -1; // Not found
264 }
```

**Listing 2.11:** StakeRouter.sol

**Impact** This can cause the `removeValidator` function to fail due to gas exhaustion if the array is too large.

**Suggestion** Traverse the `validators` array from the end.

### 2.3.2 Avoid using `address.transfer` for transferring native tokens

**Status** Fixed by [Version 2](#)

**Introduced by** [Version 1](#)

**Description** In the `iBTC` contract, the `withdrawBTC` function calls the `transfer` function to return native tokens to `msg.sender`. However, this can result in a transfer failure if the recipient is a contract and its `fallback` function contains complex logic that exceeds the gas limit imposed by transfer (i.e., 2300 gas units). It is recommended to use the `sendValue` function from OpenZeppelin's `Address` library to avoid such issues.

```
209 function withdrawBTC() external nonReentrant {
210     uint256 totalAmount = _processWithdrawals();
211
212     // Convert the XBTC to BTC using the XBTC contract's `withdraw` method
213     ixbtc.withdraw(totalAmount);
214
215     // Transfer the BTC to the user
216     payable(msg.sender).transfer(totalAmount);
217
218     emit Withdraw(msg.sender, totalAmount);
219 }
220\begin{lstlisting}[language=Solidity, caption=iBTC.sol, firstnumber=209]
221 function withdrawBTC() external nonReentrant {
222     uint256 totalAmount = _processWithdrawals();
```

```
223
224     // Convert the XBTC to BTC using the XBTC contract's `withdraw` method
225     ixbtc.withdraw(totalAmount);
226
227     // Transfer the BTC to the user
228     payable(msg.sender).transfer(totalAmount);
229
230     emit Withdraw(msg.sender, totalAmount);
231 }
```

**Listing 2.12:** iBTC.sol

**Impact** This may lead to unexpected transfer failures.

**Suggestion** Replace it with `Address.sendValue` from the OpenZeppelin library.

### 2.3.3 Remove redundant operations in the iBTC contract

**Status** Fixed by [Version 2](#)

**Introduced by** [Version 1](#)

**Description** In the `deposit`, `depositBTC`, and `requestWithdraw` functions, the initial invocation of the `_settleReward` function to handle reward settlement is redundant. This operation is already performed by the `_beforeTokenTransfer` function, which is called by the `_mint` or `_burn` functions.

```
77     function _beforeTokenTransfer(address from, address to, uint256 amount) internal override {
78         if (from != address(0)) {
79             _settleReward(from);
80         }
81         if (to != address(0)) {
82             _settleReward(to);
83         }
84         super._beforeTokenTransfer(from, to, amount);
85     }
```

**Listing 2.13:** iBTC.sol

```
110     function deposit(uint256 amount) public nonReentrant {
111         _settleReward(msg.sender);
112
113         require(amount > 0, "Amount must be greater than 0");
114         xbtc.safeTransferFrom(msg.sender, address(this), amount);
115
116         // Approve and stake the deposited XBTC with the StakeRouter
117         xbtc.safeApprove(address(stakeRouter), amount);
118         stakeRouter.deposit(amount);
119
120         // Mint iBTC tokens at a 1:1 ratio with XBTC
121         _mint(msg.sender, amount);
122
123         // Update user's reward debt based on their new balance
124         rewardDebt[msg.sender] = (balanceOf(msg.sender) * accRewardPerShare) / PRECISION;
```

```
125     emit Deposit(msg.sender, amount);
126 }
127
128 function depositBTC() public payable nonReentrant {
129     uint256 amount = msg.value;
130     require(amount > 0, "Amount must be greater than 0");
131     ixbtc.deposit{value: amount}();
132
133     _settleReward(msg.sender);
134
135     // Approve and stake the deposited XBTC with the StakeRouter
136     xbtc.safeApprove(address(stakeRouter), amount);
137     stakeRouter.deposit(amount);
138
139     // Mint iBTC tokens at a 1:1 ratio with XBTC
140     _mint(msg.sender, amount);
141
142     // Update user's reward debt based on their new balance
143     rewardDebt[msg.sender] = (balanceOf(msg.sender) * accRewardPerShare) / PRECISION;
144     emit Deposit(msg.sender, amount);
145 }
146
147 // Allows users to request the withdrawal of XBTC by burning iBTC tokens
148 function requestWithdraw(uint256 amount) external nonReentrant {
149     uint256 userBalance = balanceOf(msg.sender);
150     require(userBalance >= amount, "Insufficient iBTC balance");
151     require(amount > 0, "Amount must be greater than zero");
152
153     _settleReward(msg.sender);
154
155     // Burn the corresponding amount of iBTC tokens
156     _burn(msg.sender, amount);
157
158     // Update user's reward debt to reflect their new balance
159     rewardDebt[msg.sender] = (balanceOf(msg.sender) * accRewardPerShare) / PRECISION;
160
161     // Create a new withdrawal request with an unlock timestamp determined by the StakeRouter
162     uint256 unlockTimestamp = block.timestamp + stakeRouter.lockTime();
163     userWithdrawals[msg.sender].push(WithdrawalRequest({
164         amount: amount,
165         unlockTimestamp: unlockTimestamp
166     }));
167
168     // Initiate withdrawal with the StakeRouter
169     stakeRouter.withdraw(amount);
170
171     emit WithdrawRequested(msg.sender, amount, unlockTimestamp);
172 }
```

**Listing 2.14:** iBTC.sol

**Impact** Redundant operations can lead to more gas consumption.

**Suggestion** Remove redundant operations.

### 2.3.4 Optimize the sorting algorithm to reduce gas consumption

**Status** Fixed by [Version 3](#)

**Introduced by** [Version 2](#)

**Description** In the [StakeRouter](#) contract, the [updateValidator](#) function sorts validators by priority using the [\\_sortUpdatedValidatorsByPriority](#) function after reconfiguring a validator. The [\\_sortUpdatedValidatorsByPriority](#) function implements bubble sort, which has high computational complexity and leads to excessive gas usage. Notably, the priorities of the [validators](#) array from 0 to [index](#)-1 and from [index](#)+1 to [validators.length](#)-1 are already ordered. Therefore, only one side needs to be traversed during the sorting process.

```

88  function _sortUpdatedValidatorsByPriority() internal {
89      uint256 n = validators.length;
90
91      for (uint256 i = 1; i < n; i++) {
92          Validator memory current = validators[i];
93          uint256 j = i;
94
95          while (j > 0 && validators[j - 1].priority < current.priority) {
96              validators[j] = validators[j - 1];
97              j--;
98          }
99
100         validators[j] = current;
101     }
102 }

```

**Listing 2.15:** StakeRouter.sol

```

122  function updateValidator(
123      address _validator,
124      uint256 _minStakePerTx,
125      uint256 _maxStake,
126      uint256 _priority
127  ) external onlyOperator {
128      require(_minStakePerTx <= _maxStake, "Minimum stake must be less than maximum stake");
129
130      int256 index = getValidatorIndex(_validator);
131      require(index >= 0, "Validator not found");
132
133      Validator storage validator = validators[uint256(index)];
134
135      if (validator.currentStake > 0) {
136          require(
137              validator.currentStake <= _maxStake,
138              "Current stake must be less than or equal to the maximum stake"
139          );
140      }
141
142      validator.minStakePerTx = _minStakePerTx;
143      validator.maxStake = _maxStake;
144      validator.priority = _priority;

```



```
145
146     emit ValidatorUpdated(_validator, _minStakePerTx, _maxStake, _priority);
147     _sortUpdatedValidatorsByPriority();
148 }
```

**Listing 2.16:** StakeRouter.sol

**Impact** May lead to excessive gas consumption.

**Suggestion** Optimize the sorting algorithm.

## 2.4 Note

### 2.4.1 Potential centralization risk

**Introduced by** [Version 1](#)

**Description** The protocol includes several privileged functions that modify critical configurations, such as upgrading the implementation, transferring [XSAT](#) in the [WIBTC](#) contract, and configuring validators in the [StakeRouter](#) contract. If the private key of a privileged role is lost or maliciously exploited, it could lead to significant losses for users.

